MATHEMATICS 201-NYC-05
Vectors and Matrices
Martin Huard
Fall 2007

Systems of Linear Equations with Maple

As with matrices, we start by loading the *LinearAlgebra* package.

> **with(LinearAlgebra);**

Let us consider the following system of linear equations $AX = b$.

$$\begin{aligned} 3x + 3y + 2z &= 3 \\ 4x - 2y + 5z &= -15 \\ 2x - y - z &= 3 \end{aligned}$$

We will see five different ways to solve this system of linear equations with Maple, namely *directly*, using *Gaussian elimination*, *Gauss-Jordan method*, the *inverse* and with *Cramer's Rule*.

Solving Directly

We can solve our system of linear equations directly using the `solve()` command.

> **solve({ 3*x+3*y+2*z=3, 4*x-2*y+5*z=-15, 2*x-y-z=3 }, {x,y,z});**

Note that this method works for <u>any</u> kind of equations, not just systems of linear equations.

There is an other way to solve this using the `LinearSolve()` command from the *LinearAlgebra* package.

We begin by defining the matrix of coefficients *A* along with the constant matrix *b*.

> **A:=Matrix([[3,3,2],[4,-2,5],[2,-1,-1]]);**

> **b:=Matrix([[-3],[-15],[3]]);**

Then, the solution to our system of linear equations is:

> **LinearSolve(A,b);**

Gaussian Elimination

The first thing we need to do is find the augmented matrix. We can either rewrite the matrix, or, since we already defined *A* and *b*, form a new matrix with them.

> **C:=Matrix([A,b]);**

Now we can reduce our matrix using the `GaussianElimination( )` command from the *LinearAlgebra* package.

> **G:=GaussianElimination(C);**

Note: Maple does not give the matrix in row-echelon form since it does not have leading ones.

The solution is obtained with the `BackwardSubstitute( )` command.

> **BackwardSubstitute(G);**

Reducing the Matrix Row by Row

Instead of using the `GaussianElimination()` command, we can obtain the reduced matrix by transforming the augmented matrix through the use of elementary row operations.

The command for elementary row operations is `RowOperation`, where you can use the option `inplace=true` if you want to overwrite the matrix, so as not to have to rename it everytime.

1. `RowOperation(A,[ri,rj],m)` $R_i \to R_i + mR_j$

2. `RowOperation(A,ri,k)` $R_i \to kR_i$
3. `RowOperation(A,[ri,rj])` $R_i \leftrightarrow R_j$

Let us reduce our augmented matrix C with these elementary row operations.
Let us start with $R_1 \to \frac{1}{3}R_1$ to make a leading 1.

> **RowOperation(C,1,1/3,inplace=true);**

and make a zeros under the leading 1, with $R_2 \to R_2 - 4R_1$ and $R_3 \to R_3 - 2R_1$

> **RowOperation(C,[2,1],-4,inplace=true);**
> **RowOperation(C,[3,1],-2,inplace=true);**

Making a leading one where the -6 is with $R_2 \to -\frac{1}{6}R_2$

> **RowOperation(C,2,-1/6,inplace=true);**

and a zero under that new leading one $R_3 \to R_3 + 3R_2$

> **RowOperation(C,[3,2],3,inplace=true);**

Lastly, making a leading one where the $\frac{-7}{2}$ with $R_3 \to -\frac{2}{7}R_3$

> **RowOperation(C,3,-2/7,inplace=true);**

Which gives us the row-echelon form for our augmented matrix.

Gauss-Jordan Method

The Gauss-Jordan method works the same way for Gaussian elimination, except that we use the command `ReducedRowEchelonForm( )` from the *LinearAlgebra* package.

> **ReducedRowEchelonForm(C);**

This is probably the best method to use as it gives us a matrix where the solution stands out.

Using the Inverse

In solving the system $AX = b$, we can solve for *X* by finding the inverse of *A*, if *A* is invertible, then multiplying it by b. That is, $X = A^{-1}b$.

> **Multiply(MatrixInverse(A),b);**

Note that this method will only work if our system has a unique solution.

Cramer's Rule

To use Cramer's rule, we need to find the matrices *A*(1), *A*(2) and *A*(3), where the matrix *A*(j) is the matrix *A* with the j^{th} column replaced by the *b*. This can be done as follows.

> **A1 := Matrix([b, Column(A,2), Column(A,3)]);**
> **A2 := Matrix([Column(A,1), b, Column(A,3)]);**
> **A3 := Matrix([Column(A,1), Column(A,2) , b]);**

With these matrices, the solution is obtained with $\frac{\det(A(j))}{\det(A)}$.

> **Determinant(A1)/Determinant(A);**
> **Determinant(A2)/Determinant(A);**
> **Determinant(A3)/Determinant(A);**

Maple Tutor

Maple has a number of built-in tutors, such as Gaussian elimination, that you can explore. Go to TOOLS – TUTORS – LINEAR ALGEBRA and pick the tutor you wish to try.